
胎压计说明书－HT46R71D 

文件编码：HA0105S  

简介 
本装置是一个以 HT46R71D 为主芯片，配合传感器组成的汽车轮胎气压检测装

置，最终取高三位有效值于 LCD 显示，有四种显示单位供切换（Psi、Bar、Kpa、
Kgf/cm2）。本装置仅为参考范例，使用时，为提高准确度，建议多量测些组数

据来构建 Sensor Output Voltage - Discharge Time (V-T)表格，并作相应校准。 
 

芯片规格 
• 使用 MCU 母体：HT46R71D  
• MCU 工作电压：3.0V  
• 工作频率：4MHz（RC） 
 
按键 

1 个单位切换设置按键： SELECT 
 

控制 IO 口 
1 个控制 I/O 口：高电位执行的 IO 口（驱动工作指示灯）。 

功能描述 
• MCU 未上电或休眠状态下，接通电源或按‘SELECT’键，叫醒 MCU，

系统点亮工作指示灯，同时测量空载气压值，并保存此校准值（校准值需

经计算后以 Sensor 输出电压值形式保存）。 
• 若无按键，则取 10 次放电时间的平均值作为此次测量的 Tc 值（放电时间

Tc 1S 钟刷新测量一次），查表计算，并进行校准，取运算结果的高三位有

效值于 LCD 显示；若检测到按键发生，则进入相应的按键处理流程。 
• 若显示数据大于 999，则显示“---”，表示测试值超出显示的数值范围。 
• MCU 工作状态下，每按一次‘SELECT’键，变换一种显示单位。显示单

位按 Psi=>Bar=>Kpa=>Kgf/cm2顺序循环变换。单位变换同时，显示数值 
也作相应改变（单位之间的转换关系请参阅附录）。 

• MCU 工作状态下，若 60S 没有检测到按键，则关闭 LCD 显示，置 0 驱动

LED 的 IO 口，MCU 进入睡眠状态。 

• 某些参数的定义，要根据Sensor和所建的 Sensor Output Voltage - Discharge 
Time (V-T)表格作相应修改，如下说明： 

#DEFINE   PER                                     100   

#DEFINE   PSI_FULL                          10000   

#DEFINE   BAR_FULL                      689   

#DEFINE   KPA_FULL_L16                 0D54H  

#DEFINE   KPA_FULL_H16                01H   

#DEFINE   KGFCM2_FULL                 703   

#DEFINE  TABLE_START_ADDR          0710H   

#DEFINE   TABLE_NUM                   13   

#DEFINE   TABLE_PER                    12   

_____________________________________________________________________ 

注意:    PER:                    Sensor 满量程值放大 100 倍 
PSI_FULL：               Sensor 满量程值 100(PSI)*100=10000 
BAR_FULL：            Sensor 满量程值 6.89(PSI)*100=6.89 
KPA_FULL_L16：       Sensor满量程值 689.48(PSI)*100=68948 的低 16位 
KPA_FULL_H16：       Sensor满量程值 689.48(PSI)*100=68948 的高 16位 
KGFCM2_FULL：       Sensor 满量程值 7.03(PSI)*100=703 
TABLE_START_ADDR：  V-T 表格首地址 
TABLE_NUM：           V-T 表格数据个数 
TABLE_PER：          本装置 V-T 表格中将 Sensor 输出分为 12 份，即每 

隔 5mV 建立一组数据 
           ______________________________________________________________________ 

应用电路 
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

图  一

工作原理 
HT46R71D 为双积分模/数转换型单片机，芯片内含有放大器，电压跟随器，

积分器和比较器。充电时，内部多路开关切换到放大器输出端，经由积分电路

为充放电电容 Vc 充电；放电时，多路开关切换到 VDSO，Vc 即开始放电，当

其电压降至 1/6VDSO 时，比较器即输出低电平，视为放电结束。 

HT46R71D 的内建 3.3V 基准电压可作为 Sensor 的电压源，VOBGP PIN 可为

Sensor 提供 1.5V 的参考电压。Sensor 的输出电压需经 Amplifier 放大，Amplifier
模块的连接方式具体如下（图二）： 

 

 

 

 

 

 

 

 

 

 

 

 

 
其中，VDOPAO = VOBGP+(VA-VB)*(R2/R1)  

= 1.5V+10*(VA-VB)  (and R1=R3, R2=R4 ) 

本装置所选用 Sensor 规格为：3.3V 工作电压下满量程（100PSI）输出 60mV。 
V-T 表格以 0710 为首地址，以（Sensor 输出电压/5mV+10H）作为偏移地址，

共包含 13 组数据。在量测轮胎气压时，与所制表格数剧相同充电时间下，每 
量测一组放电时间，即可查表来确定与其值相近的表格偏移地址和数值，计算

后，Sensor 输出电压值需取至小数点后两位，以增加准确度。再经单位转换，

取运算结果的高三位有效值于 LCD 显示。 

注意事项 
• LCD 显示界面： 
 
 
 
 
 
 
 
 

设其自左往右的排布为：数据位 1，Dot1，数据位 2，Dot2，数据位三，

并列的四个转换单位。则其与程序相对应的分布如下表： 
 
 
 
 
 
 

• 请注意要根据所选用 Sensor 的规格来设置 Amplifier 的放大倍数，Sensor
输出信号经 Amplifier 放大后，应不超过 4/6VDSO 。 

• 要选用合适的充放电电阻 Vr，电容 Vc 和充电时间 Ti，使 Vc 电压介于

1/6VDSO 和 5/6VDSO 之间。 
• 充电时间改变，Regular 3.3V 稳压输出不同等因素都会影响到放电时间值，

应注意 Sensor Output Voltage - Discharge Time (V-T)表格作相应校准。为增

加准确度，也建议使用时多量测些组数据来构建表格。 

Pin  Seg0 Seg1 Seg2 Seg3 Seg4 Seg5 Seg6 Seg7 Seg8 

Com0 1f 1a Bar 2f 2a Kpa 3f 3a Kgf/cm2 

Com1 1e 1g 1b 2e 2g 2b 3e 3g 3b 

Com2 1d 1c Dot1 2d 2c Dot2 3d 3c Psi 

参考流程 
START

CALL INI_RAM

CALL INI_IO

CALL SBR_KEY_SCAN

CALL SBR_KEY_JUMP

F_HALT=0?

CLR LED_IO，LED灯熄灭

HALT

CLR F_HALT, 设置唤醒后各寄存器和IO口状态

Y

N

设置各相关寄存器

CCALL SBR_ADC,测固定充电时间条件下的放电时间

进行十次充放电，取其平均值保存于[R_TMR1H,R_TMR1L]

CALL SBR_TABLE，查表得放电时间所对应的偏移地址和相关表格值

CALL SBR_CALT，计算得二进制压力值并保存

CALL SBR_DIS, LCD 显示

1S计时到?

SET F_ON

NY

SBR_TABLE

[R_TMR1H,R_TMR1L]>
[R_TABLE1_H,R_TABLE1_H]?

分别保存V-T表格的初始地址偏移量和末地址偏移量于R_ADDR1,R_ADDR2

保存该偏移地址下的表格值于[R_TABLE1_H,R_TABLE1_L]和[R_TALBLE2_H,R_TABLE2_L]

RET

分别保存V-T表格的初始地址偏移量和末地址偏移量于R_ADDR1,R_ADDR2

设置标志位或校准值

[R_TMR1H,R_TMR1L]<
[R_TABLE2_H,R_TABLE2_H]?

取R_ADDR1与R_ADDR2的中间地址保存于ACC

ACC=0?
Y

ACC->TBLP,查表表格值低八位保存于R_TABLE_L

Y 

N

Y

N

N

TBLH=R_TMR1H?

TBLH>R_TMR1H?
R_TABLE_L=

R_TMR1L?

R_TABLE_L>
R_TMR1L?

TBLP->R_ADDR2
R_TABLE_L->R_TABLE2_L

TBLH->R_TABLE2_H

TBLP->R_ADDR1

设置标志位
TBLP->R_ADDR1

R_TABLE_L->R_TABLE1_L
TBLH->R_TABLE1_H

A

A

设置标志位或校准值

RET

RET

Y N

Y

N

N

Y

Y

N

SBR_CALT

F_ON=1?

{[R_TABLE1_H,R_TABLE1_L] - [R_TMR1_H,R_TMR1_L}*100

{[R_TABLE1_H,R_TABLE1_L] - [R_TMR1_H,R_TMR1_L}*100/
(R_TABLE1-R_TABLE2)

F_ZERO_OUTPUT=1?

F_FULL_OUTPUT=1?

压力值为0.00

RET

R_ADDR1-R_ADJUST_H
->R_ADDR1

CLR R_TEMP0
CLR R_TEMP1
CLR R_TEMP2

将结果保存为校准值 CLR F_ON
TO0-R_ADJUST_L

->DATA0

C=1?

R_ADDR1-1
R_ADDR1-

R_ADJUST_H>1?

压力值为0.00

RET

R_ADDR1-
R_ADJUST_H>0?

压力值为0.00

RET

计算小数部分放大一百倍后与
各显示单位满量程数值的乘积

上述结果/100，得小数部分与各
显示单位满量程数值的乘积

将所得成绩高到低位依次保存于
[R_TEMP2， R_TEMP1，R_TEMP0]

计算整数部分与各显示单位
满量程数值的乘积

将小数和整数部分结果累加保存于
[R_TEMP2， R_TEMP1，R_TEMP0]

B

Y

N

Y N

YN

N Y

N
Y

Y N

所得值>999 ?

999>所得值>100 ?

100>所得值>10 ?

10>所得值>1 ?

所得值<1 ?

LCD显示'- - -'

设置小数标志位和寄存器

DATA0<-TO0,DATA1<-TO1

设置小数标志位
[R_TEMP2,  R_TEMP1,

R_TEMP0]/10-
>[DATA2,DATA1,DATA0]

设置小数标志位和寄存器

设置小数标志位
[DATA2,DATA1,DATA0]

/TABLE_PER

CALL DA999,将十六进制书转换为十进制数

R_LCD3<- TO0
R_LCD2<- TO1
R_LCD1<- TO2

RET

RET

Y

N

Y

N

Y

N

Y

N

Y

N

B

[R_TEMP2,  R_TEMP1, R_TEMP0]
/100/TABLE_PER

SBR_KEY_SCAN

关timer0,清timer0标志位

4ms定时结束？

是否有键被按下？

标志位F_KEY置1

Y

N

Y

消抖计数单元减1

F_KEY=F_KEY_TEM?

消抖计数单元=0？

F_KEY->F_KEY_TEM,
还原消抖计数值

N

Y

N

F_KEY=F_KEY_PREV ?

SET F_REPEAT

F_KEY->F_KEY_PREV

CLR F_REPEAT

RET

N

Y

Y

N

为消抖计数单元赋初值

timer0寄存器初始化，开timer

标志位F_KEY清0

SBR_KEY_JUMP

60S计时加1

F_KEY=0 ?Y N

60计时结束？

标志位F_HALT置1

RET

CLR R_60S_L
CLR R_60S_H

F_REPEAT=0 ?

RET

单位转换，设置标志
位，但不能影响共用单

元的其它标志位

Y

N

 

SBR_ADC

ADCMPO=0?

寄存器设置

预充电，使充电电容充满电

RET

放电

设置相同的充电时间，charge

discharge

保存放电时间于[R_TMR1H，R_TMR1L]

N

Y

SBR_DIS

RET

R_LCD1->TBLP
SET  BP.0
40H->MP1

CALL SBR_DIS_LOOP

点亮/熄灭'BAR'&第一个小数点

R_LCD2->TBLP
MP1+1

CALL SBR_DIS_LOOP

显示/熄灭'KPA'&第二个小数点

R_LCD3->TBLP
MP1+1

CALL SBR_DIS_LOOP

显示/熄灭'KGF/CM2'&'PSI'

SBR_DIS_LOOP

读取表格值

显示数码管‘D,E,F’段

MP1+1

显示数码管‘C,G,A’段

MP1+1

显示数码管‘B’段

RET

参考程序 
 
;;******************File Name: tire gauge .Asm****************** 
;;HT_IDE Version:    V6.6 SP3 
;;Program Name:    HT46R71D.Prj 
;;Main Function:    **tire gauge ** 
;;MCU Body:    Ht46R71D 
;;VDD:      3V 
;;MCU Frequency:    4MHz 
;;------------------------------------------------------------ 
;;Mask Option: 
;;----------------------------------------------------------- 
;Osc:      RC 
;Fsys:      4MHz 
;Wake_Up PA.0-7:    PA.0 Wake_Up 
;       PA.0-PA.6 Non Wake_Up 
;Pull-high PA:     PULL-HIGH 
;Pull-high PB:     PULL-HIGH 
;Pull-high PC:     PULL-HIGH 
;Pull-high PD:    PULL-HIGH 
;PA Buzzer Function:  BZ/BZB Disable 
;Clock Source:    T1 
;Wdt:      Enable 
;WDT Clock Source:   T1 
;CLR WDT:     One clear instruction 
;WDT time-out period:   2^12/fs-2^13/fs 
;LVR:      Disable 
;LVD:      Disable 
;Int Function:    Disable 
;LCD Driver Clock:   IRCOSC/3 
;LCD ON/OFF At HALT:  LCD OFF At HALT 
;LCD Duty:     1/3 duty(3 com) 
;LCD Bias:     1/2 bias 
;;================================================ 
#INCLUDE HT46R71D.INC 
;;================================================ 
#DEFINE  R_20MS    5   ;;4mS*5=20mS 
#DEFINE  PER     100   ;;***************************** 
#DEFINE  PSI_FULL   10000  ;;*****  100(PSI)   ***** 
#DEFINE  BAR_FULL   689   ;;*****  = 6.89(BAR)  ***** 
#DEFINE  KPA_FULL_L16  0D54H  ;;*****  = 689.48(KPA)  ***** 
#DEFINE  KPA_FULL_H16  01H   ;;*****  = 7.03(KGF/CM2) ***** 
#DEFINE  KGFCM2_FULL  703   ;;***************************** 
#DEFINE      TABLE_START_ADDR 0710H   

;;the start address of Voltage-Time table    
#DEFINE  TABLE_NUM   13   ;;the Voltage-Time table number 
#DEFINE  TABLE_PER   12   ;;for calculating 
;;===================================== 
#INCLUDE MACRO.ASM 
;;--------------------------------------------- 
MAIN_DATA.SECTION 'DATA' 
R_ADDR1   EQU  [20H]   ;;to save table address 
R_ADDR2   EQU  [21H]   ;;to save table address   
R_TMR1H   EQU  [22H]   ;;to save high byte of discharging time 
R_TMR1L   EQU  [23H]   ;;to save low byte of discharge time 
R_TABLE_L   EQU  [24H]   ;;to save table value 
R_TABLE1_H   EQU  [25H]   ;;to save table value 
R_TABLE1_L   EQU  [26H]   ;;to save table value 

R_TABLE2_H   EQU  [27H]   ;;to save table value 
R_TABLE2_L   EQU  [28H]   ;;to save table value 
R_60S_L    EQU  [2DH]   ;;for timing 60s(low byte) 
R_60S_H    EQU  [2EH]   ;;for timing 60s(high byte)  
R_1S    EQU  [2FH]   ;;for timing 1s 
R_TEMP0   EQU  [24H]   ;;for calculating 
R_TEMP1   EQU  [25H]   ;;for calculating 
 
R_TEMP2   EQU  [26H]   ;;for calculating 
R_TO0    EQU  [27H]   ;;for calculating   
R_TO1    EQU  [28H]   ;;for calculating 
R_ADJUST_H   EQU  [29H]   ;;to save the high byte of discharge time for  
adjusting  
R_ADJUST_L   EQU  [2AH]   ;;to save the low byte of discharge time for  
adjusting  
F_TABLE_EQU  EQU  [2BH].0   ;;for calculating 
;F_DIS_EQU   EQU  [2BH].1   ;;for calculating 
F_HALT    EQU  [2BH].2   ;;the flag of halt mode 
;;================================================== 
KEY_DATA.SECTION 'DATA' 
DEBOUNCE_COUNTER EQU [3FH]   ;;for key detecting  
R_UNIT    EQU  [2CH]   ;;to save the unit to display 
F_REPEAT   EQU  [2BH].3   ;;the flag of key repeat 
F_KEY    EQU  [2BH].4   ;;the flag of key ok/no 
F_KEY_PREV   EQU  [2BH].5   ;;the flag of previous key 
F_ON    EQU  R_UNIT.6  ;;the flag of power on or wake up  
F_KEY_TMP   EQU  R_UNIT.7  ;;the flag for key detecting  
F_ZERO_OUTPUT  EQU  R_UNIT.5  ;;the flag of zero output of sensor 
F_FULL_OUTPUT  EQU  R_UNIT.4  ;;the flag of full output of sensor  
;;================================================= 
DISPLAY_DATA.SECTION 'DATA' 
R_LCD1    EQU  [24H]   ;;to save value displayed(decimal) 
R_LCD2    EQU  [25H]   ;;.. 
R_LCD3    EQU  [26H]   ;;.. 
R_DIS    EQU  [3EH]   ;;for display 
F_LCD_DOT1   EQU  [2BH].6   ;;the flag of dot 1  
F_LCD_DOT2   EQU  [2BH].7   ;;the flag of dot 2   
F_PSI    EQU  R_UNIT.0  ;;flag of unit (psi)  
F_BAR    EQU  R_UNIT.1  ;;flag of unit (bar)  
F_KPA    EQU  R_UNIT.2  ;;flag of unit (kpa)  
F_KGFCM2   EQU  R_UNIT.3  ;;flag of unit (kgfcm2)  
;;================================================= 
CALCULATE_DATA.SECTION 'DATA'   ;;for the calculation subroutine 
DATA0    EQU  [21H] 
DATA1    EQU  [30H] 
DATA2    EQU  [31H] 
DATA4    EQU  [32H] 
DATA5    EQU  [33H] 
DATA6    EQU  [34H] 
TO0     EQU  [35H] 
TO1     EQU  [36H] 
TO2     EQU  [37H] 
TO3     EQU  [38H] 
TO4     EQU  [39H] 
TO5     EQU  [3AH] 
TO6     EQU  [3BH] 
COUNT0    EQU  [3CH] 
COUNT1    EQU  [3DH] 
COM3    EQU  [3EH] 
COM4    EQU  [3FH] 

;;================================================= 
KEY_IO_C   EQU  PAC.0  ;;IO connecting with key 
KEY_IO    EQU  PA.0  
LED_IO_C   EQU  PAC.1  ;;IO connecting with led 
LED_IO    EQU  PA.1 
;;------------------------------------------------- 
MAIN_CODE.SECTION AT 0 'CODE' 
 
ORG  00H 
JMP  MAIN 
 
ORG  04H       ;;external int interrupt vector 
 
JMP  ISR_EXTINT 
 
ORG  08H       ;;timer0 interrupt vector 
JMP  ISR_TMR0 
 
ORG  0CH       ;;timer1 interrupt vector 
JMP  ISR_TMR1 
 
ORG  10H       ;;AD interrupt vector 
JMP  ISR_ADC 
 
;;---------------------------------------- ----------------------- 
MAIN: 
 CALL INI_RAM      ;;initial RAM  
 CALL INI_IO      ;;initial IO 
 SET  F_ON      ;;set the flag of power on/wake up 
MAIN_LOOP1:       ;;test the discharge time 
 CLR  WDT       
 CLR  R_1S 
 CLR  DATA0 
 CLR  DATA1 
 CLR  DATA2 
 XMOV R_TO0, 10     ;;charge-discharge for 10 times 
MAIN_LOOP2:       ;;calculate the average value of discharge time 
 CALL SBR_ADC     ;;test the discharge time Tc during the charge time Ti  

;;be fixed 
 CLR  C 
 XADDM DATA0, R_TMR1L 
 XADCM DATA1, R_TMR1H 
 CLR  R_TMR1H 
 XADCM DATA2,  R_TMR1H 
 SDZ  R_TO0 
 JMP  MAIN_LOOP2     ;;calculate the summation value of discharge time 
 XMOV DATA4, 10 
 CLR  DATA5 
 CLR  DATA6 
 CALL UNBIN_DIV_24    ;;calculate the average value for 10 times 
 XMOV R_TMR1L, TO0    ;;save low byte of the discharge time to R_TMR1L 
  XMOV R_TMR1H, TO1    ;;save high byte of the discharge time to R_TMR1H 
MAIN_LOOP3:       ;;calulate and display 
 CALL SBR_TABLE     ;;to get the opposite offset address of table with 
dichotomy  
 CALL SBR_CALT     ;;calculate the pressure value with different unit  
 CALL SBR_DIS      ;;display  
 CLR  WDT 
 SZ  F_HALT      ;;if at the mode of halt 
 JMP  L_HALT      ;;jump to the halt mode       

CALL SBR_KEY_SCAN    ;;detect key 

 CALL SBR_KEY_JUMP    ;;deal with key 
 INC  R_1S      
 NEJMP R_1S, 50      ;;1S timing  
 JMP  MAIN_LOOP1     ;;jump to test the discharge time     
 JMP  MAIN_LOOP3     ;;jump to calulate and display 
L_HALT:             
 CLR  LED_IO 
 HALT        ;;halt 
 CLR  F_HALT      ;;wake up 
 SET  F_KEY_PREV     ;;set the previous key flag 
 XMOV CHPRC, 63H     ;;set 3.3V regular output 
 XMOV ADCR, 41H     ;;set the related registor 
 XMOV ADCD, 07H 
 XMOV TMR1C, 88H 
  

XMOV R_UNIT, 01H     ;;set to 'psi'   
 SET  F_ON      ;;set the flag of power on/wake up     
 SET  LED_IO 
 JMP  MAIN_LOOP1     ;;test again 
;;--------------------------------------------- 
ISR_EXTINT:        ;;no used 
 RETI 
;;============================================= 
ISR_TMR0:        ;;no used 
 RETI    
;;============================================ 
ISR_TMR1:        ;;no used 
 RETI 
;;=========================================== 
ISR_ADC:        ;;no used 
 RETI 
;;========================================== 
#INCLUDE SUBROUTINE.ASM 
 
;;******************File Name: Subroutine.Asm****************** 
;;--------------------------------------------------------- 
INI_RAM:  ;;initial ram  
;;--------------------------------------------------------- 
 CLR  BP 
 XMOV MP0, 0A0H 
 CLR  WDT 
 CLR  R0 
 INC  MP0 
 EJMP MP0, 0BFH     ;;clear bank0 ram(20~3FH) 
 JMP  $-6 
 
 XMOV BP, 01H 
 XMOV MP1, 0C0H 
 CLR  WDT  
 CLR  R1 
 INC  MP1  
 EJMP MP1, 0CFH     ;;clear bank1 ram(40~4FH) 
 JMP  $-6                            
  
 CLR  INTC0      ;;interrupt disable  
 CLR  INTC1 
 XMOV CHPRC, 63H     ;;set 3.3V regular output 
 XMOV ADCR, 41H     ;;set the related registor 
 XMOV ADCD, 07H 

XMOV TMR1C, 88H    

 XMOV R_UNIT, 01H 
 RET 
;;--------------------------------------------------------- 
INI_IO:         ;;initial IO port 
;;--------------------------------------------------------- 
 CLR  WDT 
 SET  PAC 
 SET  PBC 
 CLR  LED_IO_C 
 SET  LED_IO 
 RET 
;;--------------------------------------------------------- 
SBR_ADC:         
;;test the discharge time Tc during the charge time Ti be fixed 
;;-------------------------------------------------------- 
 CLR  TMR1C.4 
 XMOV TMR0C, 8FH     ;;32us(Fsys=4MHz)  
 XMOV TMR0, 00H 
 XMOV ADCR, 01001011B    ;;at charge mode 
 SET  TMR0C.4 
 CLR  WDT 
 SNZ  T0F       ;;charge and let Vc work over 4/6 VDSO  
 JMP  $-2 
 CLR  T0F 
 
 CLR  TMR0C.4 
 XMOV ADCR, 01001101B    ;;at discharge mode 
 CLR  WDT 
 SZ  ADCMPO       
 JMP  $-2 
 
 XMOV TMR0, 83H     
 
SBR_ADC_LOOP: 
 CLR  WDT 
 MOV A, 00000110B     ;;at charge mode 
 XORM A,ADCR   
 SET  TMR0C.4  
 CLR  TMR1C.4 
 CLR  TMR1L 
 CLR  TMR1H 
 
 CLR  WDT 
 SNZ  T0F       ;;charge for 3968 S 
 JMP  $-2 
 
 MOV A, 00000110b     ;;at discharge mode 
 XORM A, ADCR  
 SET  TMR1C.4  
 CLR  TMR0C.4 
 CLR  T0F         
 CLR  WDT 
 SZ  ADCMPO     ;;discharge 
 JMP  $-2   
 CLR  TMR1C.4 
 XMOV R_TMR1H, TMR1H 
 XMOV R_TMR1L, TMR1L    ;;save discharge time  
 RET 
;---------------------------------------------------------- 

SBR_TABLE:         
;;get the opposite offset address of table with dichotomy  
;---------------------------------------------------------- 
 CLR  WDT 
 CLR  BP 
 XMOV R_ADDR1, LOW TABLE_START_ADDR  
          ;;save the V-T table start address  
 XMOV R_ADDR2, LOW TABLE_START_ADDR+TABLE_NUM-1 
          ;;save the V-T table end address 
 XMOV TBLP, R_ADDR1    
 TABRDL R_TABLE1_L  
 XMOV R_TABLE1_H, TBLH   ;;get value of the start address   
 XMOV TBLP, R_ADDR2   
 TABRDL R_TABLE2_L  
 XMOV R_TABLE2_H, TBLH   ;;get value of the end address 
 EJMP R_TABLE1_H, R_TMR1H  ;;if discharge time(Tc)=table value of start address 
   
 JMP  $+6       ;;unequal  
 LBRJ R_TABLE1_L, R_TMR1L      
 JMP  L_RET1      ;;Tc>the table value of start address 
  

JMP  SBR_TABLE_LOOP     
 LBRJ R_TABLE1_H, R_TMR1H     
 JMP  L_RET1      ;;Tc>the table value of start address 
  
 EJMP R_TABLE2_H, R_TMR1H  ;;if discharge time(Tc)=table value of end address 
   
 JMP  $+6 
 LBRJ R_TMR1L, R_TABLE2_L   
 JMP  L_RET2      ;;Tc<the table value of end address 
 JMP  SBR_TABLE_LOOP 
 LBRJ R_TMR1H, R_TABLE2_H 
 JMP  L_RET2      ;;Tc<the table value of end address 
  
SBR_TABLE_LOOP:      ;;at the testing range of sensor 
 CLR  WDT 
 MOV A, R_ADDR2 
 SUB  A, R_ADDR1 
 CLR  C 
 RRC ACC 
 SZ  ACC 
 JMP  $+2 
 JMP  L_RET_LOOP 
 ADD A, R_ADDR1    
 MOV TBLP, A   

TABRDL R_TABLE_L     ;;get the middle address between addr1 and addr2，
and save the  

;;table value to R_TALBE_L 
L_COMPARE_H:       ;;compare the high byte of Tc with table value 
 NEJMP TBLH, R_TMR1H     
 JMP  L_COMPARE_L  
 CLR  F_TABLE_EQU 
 LBRJ TBLH, R_TMR1H 
 JMP  L_BIGGER 
 JMP  L_SMALLER 
L_COMPARE_L:       ;;compare the low byte of Tc with table value 
 EJMP R_TABLE_L, R_TMR1L    
 JMP  $+3 
 SET  F_TABLE_EQU      
 JMP  L_EQU 

LBRJ R_TABLE_L, R_TMR1L 

 JMP  L_BIGGER 
 JMP  L_SMALLER  
L_BIGGER:        ;;Tc > table value 
 XMOV R_ADDR2, TBLP 
 XMOV R_TABLE2_L, R_TABLE_L 
 XMOV R_TABLE2_H,TBLH  
 JMP  SBR_TABLE_LOOP  
L_SMALLER:        ;;Tc < table value 
 XMOV R_ADDR1, TBLP  
 XMOV R_TABLE1_L, R_TABLE_L 
 XMOV R_TABLE1_H, TBLH  
 JMP  SBR_TABLE_LOOP 
L_EQU:         ;;Tc = table value  
 SNZ  F_TABLE_EQU 
 JMP  $+3 
 XMOV R_ADDR1, TBLP 
 JMP  L_RET_LOOP 
L_RET1:         ;;zero output of the sensor 
 CLR  F_KEY_TMP 
 SET  F_ZERO_OUTPUT 
 SNZ  F_ON 
 RET 
 CLR  F_ON 
  

CLR  R_ADJUST_L 
 XMOV R_ADJUST_H, R_ADDR1  ;;set the adjust value 
 RET  
 
L_RET2:         ;;full output of the sensor 
 XMOV R_ADDR1, LOW TABLE_START_ADDR+TABLE_NUM-1 
 CLR  F_KEY_TMP 
 SET  F_FULL_OUTPUT 
 SNZ  F_ON 
 RET 
 CLR  F_ON 
 CLR  R_ADJUST_L 
 XMOV R_ADJUST_H, LOW TABLE_START_ADDR+TABLE_NUM-1 
 RET         ;;set the adjust value 
  
L_RET_LOOP: 
 CLR  F_ZERO_OUTPUT 
 CLR  F_FULL_OUTPUT 
 CLR  F_KEY_TMP 
 RET 
;;--------------------------------------------------------- 
SBR_CALT:         
;;calculate pressure with different unit  
;;--------------------------------------------------------- 
 CLR  WDT 
 SZ  F_ZERO_OUTPUT     
 JMP  L_CALT_LOOP4    ;;for zero output of sensor 
 SNZ  F_FULL_OUTPUT     
 JMP  $+8 
 MOV A, R_ADDR1     ;;for full output of sensor 
 SUB  A, R_ADJUST_H 
 MOV R_ADDR1, A 
 CLR  R_TEMP0 
 CLR  R_TEMP1 
 CLR  R_TEMP2 

JMP  L_CALT_LOOP5 

 MOV A, R_TABLE1_L   
 SUB  A, R_TMR1L 
 MOV DATA0, A  
 MOV A, R_TABLE1_H 
 SBC  A, R_TMR1H 
 MOV DATA1, A 
 CLR  DATA2 
 XMOV DATA4, 100 
 CLR  DATA5  
 CLR  DATA6 
 CALL UNBIN_MUL_24    ;;{[R_TABLE1_H,R_TABLE1_L] - 
[R_TMR1_H,R_TMR1_L]*100 
  
 XMOV DATA0, TO0 
 XMOV DATA1, TO1 
 XMOV DATA2, TO2 
   
 MOV A, R_TABLE1_L 
 SUB  A, R_TABLE2_L 
 MOV DATA4, A 
 MOV A, R_TABLE1_H 
 SBC  A, R_TABLE2_H 
 MOV DATA5, A 
 CLR  DATA6 
 CALL UNBIN_DIV_24    ;;{[R_TABLE1_H,R_TABLE1_L]  

;;-[R_TMR1_H,R_TMR1_L]} 
;;*100/(R_TABLE1-R_TABLE2) 

          ;;and get two bits of (decimal*100) 
  
 SZ  F_ON      ;;if power on/wake up 
 JMP  L_INI 
 JMP  L_CALT_LOOP1 
L_INI:         ;;get the adjust value 
 CLR  F_ON 
 XMOV R_ADJUST_L, TO0 
 XMOV R_ADJUST_H,R_ADDR1 
   
L_CALT_LOOP1:       ;;calculate pressure with different unit 
 SET  C  
 MOV A, TO0 
 SUB  A, R_ADJUST_L    ;;adjust 
 MOV  DATA0,  A 
 
 SZ  C 
 JMP  L_CALT_LOOP3     
 SET  C       ;;TO0<R_ADJUST_L 
 MOV A, R_ADDR1 
 SUB  A, R_ADJUST_H   
 SNZ  C       ;;if R_ADDR1 is bigger than R_ADJUST_H or not 
 JMP  L_CALT_LOOP4     ;;not 
 SZ  ACC       ;;if R_ADDR1-R_ADJUST_H>1? 
 JMP  $+2 
 JMP  L_CALT_LOOP4    ;;not 
 DEC  R_ADDR1     ;;because 'TO0<R_ADJUST_L' 
L_CALT_LOOP3:       ;;TO0>R_ADJUST_L 
 SET  C 
 MOV A, R_ADDR1 
 SUB  A, R_ADJUST_H   

SNZ  C        

 JMP  L_CALT_LOOP4     ;;R_ADDR1<R_ADJUST_H 
 MOV R_ADDR1, A      ;;R_ADDR1>R_ADJUST_L 
 JMP  L_CALT_LOOP2   
L_CALT_LOOP4:        ;;for zero output of sensor   
 CLR  R_LCD1       ;;to display 0.00 
 CLR  R_LCD2 
 CLR  R_LCD3 
 SET  F_LCD_DOT1 
 CLR  F_LCD_DOT2 
 RET          
 
L_CALT_LOOP2:            
 CLR  DATA1 
 CLR  DATA2 
 CALL SBR_CALT_LOOP 
 CALL UNBIN_MUL_24     ;;decimal*100*[psi/bar/kpa/kgfcm2_full] 
   
 XMOV DATA0, TO0 
 XMOV DATA1, TO1 
 XMOV DATA2, TO2 
 XMOV DATA4, 100 
 CLR  DATA5 
 CLR  DATA6 
 CALL UNBIN_DIV_24     ;;decimal*[psi/bar/kpa/kgfcm2_full] 
 
 XMOV R_TEMP0, TO0 
 XMOV R_TEMP1, TO1 
 XMOV R_TEMP2, TO2 
 
L_CALT_LOOP5:        
 XMOV DATA0, R_ADDR1                             
 CLR  DATA1 
 CLR  DATA2 
 CALL SBR_CALT_LOOP 
 CALL UNBIN_MUL_24     ;;integer*[psi/bar/kpa/kgfcm2_full] 
  
 XADDM R_TEMP0, TO0 
 XMOV DATA0, R_TEMP0 
 XADCM R_TEMP1, TO1 
 XMOV DATA1, R_TEMP1 
 XADCM R_TEMP2, TO2 
 XMOV DATA2, R_TEMP2            ;;(decimal+integer)*[psi/bar/kpa/kgfcm2_full] 
and save                             
 XMOV DATA4, PER 
 CLR  DATA5 
 CLR  DATA6 
 CALL UNBIN_DIV_24       ;;/100 
 XMOV DATA0, TO0 
 XMOV DATA1, TO1 
 XMOV DATA2, TO2 
 XMOV DATA4, TABLE_PER    ;;/TABLE_PER 
 CLR  DATA5 
 CLR  DATA6 
 CALL UNBIN_DIV_24 
    
 XMOV R_TO0, TO0 
 XMOV R_TO1, TO1 
 
 MOV A, R_TO0      ;;>999? 

SUB  A, LOW 999 

 XMOV TO1,  HIGH 999 
 MOV A, R_TO1 
 SBC  A, TO1 
 SZ  C 
 JMP  L_OVER_999 
 
 MOV A, R_TO0 
 SUB  A, 100       ;;<999 & >100? 
 MOV A, R_TO1 
 CLR  TO1 
 SBC  A, TO1 
 SZ  C 
 JMP  L_OVER_100 
 
 MOV A, R_TO0      ;;<100 & >10? 
 SUB  A, 10 
 MOV A, R_TO1 
 CLR  TO1 
 SBC  A, TO1 
 SZ  C 
 JMP  L_OVER_10 
 
 SZ  R_TO0       ;;<10 & >1? 
 JMP  L_OVER_1 
 SZ  R_TO1 
 JMP  L_OVER_1 
  
 SET  F_LCD_DOT1      ;;<1 
 CLR  F_LCD_DOT2 
 CLR  R_LCD1       ;;0.@@ 
 XMOV DATA0, R_TEMP0 
  

XMOV DATA1, R_TEMP1 
 XMOV DATA2, R_TEMP2 
 JMP  L_DEAL_LOOP1 
 
L_OVER_1:         ;;>1 & <10 
 SET  F_LCD_DOT1 
 CLR  F_LCD_DOT2      ;;@.@@ 
 XMOV DATA0, R_TEMP0 
 XMOV DATA1, R_TEMP1 
 XMOV DATA2, R_TEMP2 
 JMP  L_DEAL_LOOP1 
 
L_OVER_10:         ;;>10 & <100 
 CLR  F_LCD_DOT1      ;;@@.@ 
 SET  F_LCD_DOT2 
 XMOV DATA0, R_TEMP0  
 XMOV DATA1, R_TEMP1 
 XMOV DATA2, R_TEMP2 
 XMOV DATA4, 10 
 CLR  DATA5 
 CLR  DATA6 
 CALL UNBIN_DIV_24     
 XMOV DATA0, TO0 
 XMOV DATA1, TO1 
 XMOV DATA2, TO2 
 JMP  L_DEAL_LOOP1 

L_OVER_100:         ;;>100 
 CLR  F_LCD_DOT1 
 CLR  F_LCD_DOT2      ;;@@@ 
 XMOV DATA0, R_TO0    
 XMOV DATA1, R_TO1 
 JMP  L_DEAL_LOOP2 
 
L_OVER_999:         ;;>999 
 CLR  F_LCD_DOT1 
 CLR  F_LCD_DOT2      ;;--- 
 XMOV R_LCD3, 10 
 MOV R_LCD2, A 
 MOV R_LCD1, A 
 RET 
 
L_DEAL_LOOP1: 
 CLR  WDT 
 XMOV DATA4, TABLE_PER     
 CLR  DATA5 
 CLR  DATA6 
 CALL UNBIN_DIV_24     ;;/TABLE_PER 
 XMOV DATA0, TO0 
 XMOV DATA1, TO1 
L_DEAL_LOOP2:        ;;binary to decimal 
 CALL DA999       
 XMOV R_LCD3, TO0     
 XMOV R_LCD2, TO1 
 XMOV R_LCD1, TO2      ;;(R_LCD3,R_LCD2,R_LCD1)=@@@ 
 RET 
 
SBR_CALT_LOOP:  
 SZ  F_PSI       ;;if Psi 
 JMP  L_CALT_PSI 
 SZ  F_BAR       ;;if Bar 
 JMP  L_CALT_BAR 
  

SZ  F_KPA       ;;if Kpa 
 JMP  L_CALT_KPA 
 SZ  F_KGFCM2      ;;if Kgf/cm2 
 JMP  L_CALT_KGFCM2 
 RET  
L_CALT_PSI: 
 XMOV DATA4, LOW PSI_FULL 
 XMOV DATA5,  HIGH PSI_FULL 
 CLR  DATA6       ;;set the multiplier 
 RET 
L_CALT_BAR: 
 XMOV DATA4, LOW BAR_FULL 
 XMOV DATA5, HIGH BAR_FULL 
 CLR  DATA6       ;;set the multiplier 
 RET 
L_CALT_KPA: 
 XMOV DATA4, LOW KPA_FULL_L16 
 XMOV DATA5, HIGH KPA_FULL_L16 
 XMOV DATA6, KPA_FULL_H16   ;;set the multiplier 
 RET    
L_CALT_KGFCM2: 
 XMOV DATA4, LOW KGFCM2_FULL 
 XMOV DATA5, HIGH KGFCM2_FULL 
 CLR  DATA6       ;;set the multiplier 



                                胎压计说明书－HT46R71D 

 24  

  
RET 

;;--------------------------------------------------------- 
SBR_KEY_SCAN:        ;;detect key 
;;---------------------------------------------------------  
 XMOV DEBOUNCE_COUNTER, R_20MS ;;detect key for 20mS 
 CLR  F_KEY_TMP     
SBR_KEY_SCAN_LOOP: 
 CLR  TMR0C.4 
 XMOV TMR0, 208      ;;4mS 
 XMOV TMR0C, 0B0H 
 CLR  WDT 
 SNZ  INTC0.5 
 JMP  $-2 
 CLR  INTC0.5 
 CLR  TMR0C.4 
 
 SNZ  KEY_IO       ;;check if key press or not 
 JMP  $+3 
 CLR  F_KEY       ;;key no 
 JMP  KEY_CHECK 
 SET  F_KEY       ;;key yes 
 
KEY_CHECK:    
 CLR  WDT       ;;check if key change or not 
 SZ  F_KEY 
 JMP  $+4 
 SZ  F_KEY_TMP  
 JMP  KEY_CHANGE 
 JMP  $+2 
 SZ  F_KEY_TMP 
 JMP  $+2 
 JMP  KEY_CHANGE  
 
 DEC  DEBOUNCE_COUNTER  
 SZ  DEBOUNCE_COUNTER 
 JMP  SBR_KEY_SCAN_LOOP    
 XMOV DEBOUNCE_COUNTER, R_20MS 
 JMP  KEY_VERIFY1 
 
KEY_CHANGE: 
 SZ  F_KEY_TMP      ;;save F_KEY to F_KEY_TMP 
 JMP  $+3 
 SET  F_KEY_TMP 
 JMP  $+2  
 CLR  F_KEY_TMP 
 XMOV DEBOUNCE_COUNTER, R_20MS 
 JMP   SBR_KEY_SCAN_LOOP     
 
KEY_VERIFY1:        ;;save F_KEY to F_KEY_PREV 
 SZ  F_KEY       
 JMP  KEY_VERIFY2 
 SZ  F_KEY_PREV 
 JMP  $+3 
 SET  F_REPEAT 
 JMP  L_RET_LOOP 
 CLR  F_REPEAT 
 CLR  F_KEY_PREV 
 JMP  L_RET_LOOP      ;;ret 
KEY_VERIFY2: 
 SZ  F_KEY_PREV 

JMP  $+4 

 CLR  F_REPEAT 
 SET  F_KEY_PREV 
 JMP  L_RET_LOOP 
 SET  F_REPEAT 
 JMP  L_RET_LOOP     ;;ret 
;;--------------------------------------------------------- 
SBR_KEY_JUMP:       ;;deal with key 
;;--------------------------------------------------------- 
 CLR  WDT 
 SZ  F_KEY      ;;if key yes or no 
 JMP  L_KEY_CYCLE      
L_KEY_NO:        ;;key no        
 INC  R_60S_L        
 SNZ  Z 
 JMP  $+2 
 INC  R_60S_H  
 EJMP R_60S_L, 0B8H 
 RET 
 EJMP R_60S_H, 0BH    ;;if timing for 60s or not 
 RET         ;;if no,return  
 SET  F_HALT      ;;if yes, then set the flag of halt  
 RET 
L_KEY_CYCLE:       ;;key yes and changing the unit circularly 
 CLR  R_60S_L 
 CLR  R_60S_H  
 SZ  F_REPEAT     ;;if repeat key ,then return 
 RET 
 MOV A, R_UNIT     ;;change the unit circularly 
 AND A, 00001111B 
 MOV R_TO0, A 
 XOR A, 00001000B 
 SNZ  Z 
 JMP  $+4 
 CLR  ACC 
 SET  ACC.0 
 JMP  $+3 
 MOV A, R_TO0 
 RL  ACC 
 
 SZ  R_UNIT.4 
  

SET  ACC.4 
 SZ  R_UNIT.5 
 SET  ACC.5 
 SZ  R_UNIT.6 
 SET  ACC.6 
 MOV R_UNIT, A 
 RET 
;;--------------------------------------------------------- 
SBR_DIS:        ;;display the presure value(decimal ) 
;;---------------------------------------------------------  
 CLR  WDT      
 XMOV TBLP, R_LCD1   
 XMOV BP, 01H 
 XMOV MP1, 40H 
 CALL SBR_DIS_LOOP    ;;display R_LCD1 
 
 CLR  R1.0  
 SNZ  F_BAR      

JMP  $+2 

 SET  R1.0        ;;display 'BAR' 
 
 CLR  R1.2 
 SNZ  F_LCD_DOT1 
 JMP  $+2 
 SET  R1.2        ;;display 'dot1' 
  
 XMOV TBLP, R_LCD2        
 INC  MP1 
 CALL SBR_DIS_LOOP     ;;display R_LCD2 
 
 CLR  R1.0 
 SNZ  F_KPA 
 JMP  $+2 
 SET  R1.0        ;;display 'KPA' 
 
 CLR  R1.2 
 SNZ  F_LCD_DOT2 
 JMP  $+2 
 SET  R1.2        ;;display 'dot2' 
 
 XMOV TBLP, R_LCD3 
 INC  MP1 
 CALL SBR_DIS_LOOP     ;;display R_LCD3 
   
 CLR  R1.0 
 SNZ  F_KGFCM2 
 JMP  $+2 
 SET  R1.0        ;;display 'KGFCM2' 
  
 CLR  R1.2 
 SNZ  F_PSI 
 JMP  $+2 
 SET  R1.2        ;;display 'PSI' 
 CLR  BP 
 RET  
;;------------------------------------------------------- 
SBR_DIS_LOOP:       
 CLR  WDT 
 TABRDL R_DIS       ;;read the table value 
 XAND R_DIS, 07H 
 MOV R1, A       ;;display 'D,E,F' segment 
  
  

INC  MP1 
 SWAPA R_DIS 
 AND A, 00000111B 
 MOV R1, A       ;;display 'C,G,A' segment 
   
 INC  MP1 
 SNZ  R_DIS.3 
 JMP  $+3 
 SET  R1.1        ;;display 'B' segment 
 RET  
 CLR  R1.1 
 RET 
;;-------------------------------------------------------- 
DA999:                          ;;(DATA1,DATA0)H->(TO2,TO1,TO0)D 
;;-------------------------------------------------------- 

CLR  WDT 

 CLR  TO0 
 CLR  TO1 
 CLR  TO2 
DA_L2: 
    MOV    A, DATA0 
    SUB     A, 100 
    SZ      C 
    JMP     DA_L0 
    SZ       DATA1 
    JMP     DA_L1 
DA_L4: 

CLR  WDT 
    MOV    A, DATA0 
    SUB     A, 10 
    SZ       C 
    JMP      DA_L3 
    MOV    A, DATA0 
    MOV    TO0, A 
    RET 
DA_L3: 
 CLR  WDT 
    MOV    DATA0, A 
    INC     TO1 
    JMP     DA_L4 
DA_L1: 
    DEC     DATA1 
DA_L0: 
 CLR  WDT 
    MOV    DATA0, A 
    INC     TO2 
    JMP     DA_L2 
;========================================================= 
;24 BIT DATA CALCULATION 
;========================================================= 
UNBIN_ADD_24 PROC      
 INIT 
     MOV A, DATA0          
 ;DATA0DATA1DATA2+DATA4DATA5DATA6---->TO0TO1TO2TO3  
    ADD  A, DATA4 
    MOV TO0, A 
    MOV A, DATA1 
    ADC  A, DATA5 
    MOV TO1, A 
    MOV A, DATA2 
    ADC  A, DATA6 
    MOV TO2, A 
    MOV A, 00H 
     

ADCM A, TO3 
    RET 
UNBIN_ADD_24 ENDP 
 
;24 BIT UNSIGNED SUB 
UNBIN_SUB_24 PROC         
 ;DATA0DATA1DATA2-DATA4DATA5DATA6---->TO0TO1TO2TO3 
 INIT 
 MOV A, DATA0 
 SUB  A, DATA4 
    MOV TO0, A 

MOV A, DATA1 

 SBC  A, DATA5 
    MOV TO1, A 
    MOV A, DATA2 
 SBC  A, DATA6 
    MOV TO2, A 
 MOV A, 00H 
 SBCM A, TO3 
 RET 
UNBIN_SUB_24 ENDP 
 
;24 BIT UNSIGNED MUL  
UNBIN_MUL_24 PROC  
 ;DATA0DATA1DATA2*DATA4DATA5DATA6---->TO0TO1TO2TO3TO4TO5 
 INIT 
 MOV A, 18H 
 MOV COUNT0, A     
     CLR [0AH].0   
RRADD24: 
 CLR  WDT 
 RRC  TO5 
    RRC  TO4 
    RRC  TO3 
    RRC  DATA6 
    RRC  DATA5          
    RRC  DATA4 
    SNZ  [0AH].0       
    JMP  RR124 
    MOV A, DATA0 
    ADDM A, TO3 
    MOV A, DATA1 
    ADCM A, TO4 
    MOV A, DATA2 
    ADCM A, TO5 
RR124:  
 CLR  WDT 
  SDZ  COUNT0 
    JMP  RRADD24 
    RRC  TO5      
    RRC  TO4 
    RRC  TO3 
    RRC  DATA6 
    RRC  DATA5 
    RRC  DATA4 
    MOV A, DATA4 
    MOV TO0, A 
    MOV A, DATA5 
    MOV TO1, A 
    MOV A, DATA6 
    MOV TO2, A 
 RET 
UNBIN_MUL_24 ENDP 
 
;24  BIT UNSIGNED DIV 
UNBIN_DIV_24 
PROC               ;DATA0DATA1DATA2/DATA4DATA5DATA6---->DATA0DATA1DATA2(TO0
TO1TO2)---TO4TO5TO6   
    INIT 
    MOV    A,18H    
    MOV    COUNT0, A  



                                胎压计说明书－HT46R71D 

 29  

     
SZ      DATA6    

    JMP     START24 
    SZ      DATA5 
    JMP     START24 
    SZ      DATA4            
    JMP     START24 
    JMP     OVER24 
START24: 

CLR  WDT 
    SZ  DATA2    
    JMP     DIV24 
    SZ  DATA1 
    JMP     DIV24 
    SZ  DATA0 
    JMP     DIV24 
    JMP     DISPA24 
DIV24:   

CLR  WDT 
    CLR     [0AH].0  
    RLC     DATA0 
    RLC     DATA1 
    RLC     DATA2 
    RLC     TO4 
    RLC     TO5   
    RLC     TO6      
    MOV A, TO4   
    SUB     A, DATA4 
    MOV COM3, A 
    MOV A, TO5 
    SBC  A, DATA5 
    MOV COM4, A 
    MOV A, TO6 
    SBC     A, DATA6 
    SNZ  [0AH].0         
    JMP     NEXT24   
    MOV    TO6, A 
    MOV    A, COM3 
    MOV    TO4, A 
    MOV    A, COM4 
    MOV    TO5, A   
    MOV    A, 01H 
    ADDM  A, DATA0 
    MOV A, 00H 
    ADCM  A, DATA1 
    ADCM  A, DATA2 
NEXT24:   

CLR  WDT 
    SDZ     COUNT0 
    JMP     DIV24 
DISPA24: 

CLR  WDT 
    MOV  A, DATA0 
    MOV    TO0, A  
    MOV    A, DATA1 
    MOV    TO1, A        

MOV    A, DATA2 
    MOV    TO2, A               
    MOV    A, 00H 
    MOV    TO3, A     

RET 

OVER24: 
    RET 
UNBIN_DIV_24 ENDP 
;;--------------------------------------------------------- 
TABLE_CODE.SECTION AT 0700H 'CODE'       ;;xcga bdef 
 
 DC 05FH, 048H, 03EH, 07CH, 069H    ;;0~4;;0700H 
 DC 075H, 077H, 058H, 07FH, 07DH    ;;5~9;;0709H  
 DC 020H, 000H, 000H, 000H, 000H    ;;'-';;070EH 
 DC 000H            ;;070FH 
;;********************* V-T table************************ 
 DC 6526, 6155, 5838, 5520, 5203     ;0,5,10,15,20;;0~4mv 
 DC 4885, 4568, 4250, 3933, 3615     ;25,30,35,40,45;;5~9mv 
 DC 3298, 2980, 2662         ;50,55,60;;10~14mv   

附录 

 

 

牛顿/米 2 

(帕斯卡) 

(N/m
2
) 

(Pa) 

公斤力/

米 2 

(kgf/m2) 

公斤力/ 

厘米 2 

(kgf/cm2)

巴 

(bar) 

标准大气

压(atm)

毫米水柱 

4oC 

(mmH2O) 

毫米水银

柱 0oC 

(mmHg) 

磅/英寸
2

(lb/in2,

psi) 

牛顿/米
2
 

(帕斯卡) 

(N/m2)(Pa) 

1 0.101972 
10.1972 

×10
-6
 

1×10-5 
0.986923

×10
-5
 

0.101972 
7.50062 

×10
-3
 

145.038 

×10
-6
 

公斤力/米 2 

(kgf/m2) 
9.80665 1 1×10

-4
 

9.80665 

×10-5 

9.67841

×10-5 
1×10

-8
 0.0735559 0.00142233

公斤力/厘

米
2
 

(kgf/cm2) 

98.0665 

×103 
1×10

4
 1 0.980665 0.967841 10×10

3
 735.559 14.2233 

巴 

(bar) 
1×105 10197.2 1.01972 1 0.986923

10.1972 

×103 
750.061 14.5038

标准大气压 

(atm) 

1.01325 

×105 
10332.3 1.03323 1.01325 1 

10.3323 

×103 
760 14.6959 

毫米水柱 

4oC 

(mmH2O) 

0.101972 1×10-8 1×10-4 
9.80665 

×10-5 

9.67841

×10-5 
1 

73.5559 

×10-3 

1.42233 

×10-3 

毫米水银柱 

0oC 

(mmHg) 

133.322 13.5951 0.00135951 0.00133322
0.0013157

9 
13.5951 1 0.0193368

磅/英寸 2 

(lb/in2, 

psi) 

6.89476 

×103 
703.072 0.0703072 0.0689476 0.0680462 703.072 51.7151 1 

注：1. 1 工程大气压(at)=1 公斤力/厘米 2. 

xxxx2. 用水柱表示的压力，是以纯水在 4
o
C 时的密度值为标准的. 


